

# מחשבת 7

התוכנית הגדולה

commodore64

## משהו על חברת "מחשבת" – מערכות למידה

זו חברה, אשר אחד מתחומי התמחותה העיקריים הוא פיתוח מערכות לימוד עצמי.

מחברי יחידות לימוד אלה עוסקים במשך שנים בפיתוח מערכות ללימוד עצמי עבור המגזר האזרחי והצבאי.

חל איסור מוחלט להעתיק או לשכפל יחידת לימוד זו בשלמותה או בחלקה לכל מטרה שהיא או לעשות בה שימוש מסחרי כלשהו, ללא רשות בכתב מאת חברת מחשבת – מ.ל. בע"מ.

© 1986 – כל הזכויות שמורות למחשבת – מ.ל. בע"מ.

ת.ד. 48032 ת"א מיקוד 61480.

הוצאה לאור – מחשבת מ.ל. בע"מ.

**Copyright © 1986 by MAKH-SHEVET M.L. Ltd.  
P.O.Box 48032 Tel-Aviv 61480, Israel**

COMMODORE 64

**BUG**

MICROCOMPUTERS BOOKS & SOFTWARE

המרכז לספרות עזר ותוכנה למחשבים

# מחשבת

צעדים מתקדמים בתכנות  
בלימוד עצמי

יחידה 7  
התוכנית הגדולה

## עמוד

5	..... פתח דבר
6	..... פרק א - אותיות קידוש לבנה
21	..... פרק ב - משחקי מספרים
27	..... פרק ג - התוכנית הגדולה - משחקי מילים
36	..... פרק ד - מחסנית המקשים
44	..... פרק ה - מוט ההגוי - JOYSTICK
51	..... פרק ו - ירי בצלחות חרס
53	..... תשובות
66	..... רשימת מושגים נלמדים

יחידה 7, "התוכנית הגדולה", מוקדשת לפיתוח תוכניות מגוונות מתחומים שונים, תוך שימוש בכל הכלים התכנותיים אשר כבר למדת בשש חוברות "מחשבת" הקודמות עבור הקומודור 64.

- כידוע לך שפת ה-BASIC של הקומודור 64 עשירה ביותר, ועדיין לא לימדנו את כל אשר היא מקיפה. כהרגלנו, נמשיך ונלמד אותך מושגים חדשים שיתוספו למאגר הידע הרחב שכבר צברת. בין השאר:
  - נטפל בצורה מתוככמת במחרוזות ותוך כדי כך תלמד פקודות חדשות לטיפול במחרוזות.
  - נחשוף את סוד פעולתה של מחסנית-המקשים על מנת לפתח "משחק-מספרים" מתוככם.
  - נלמד כיצד לרתום את הג'ויסטיק (JOYSTICK) לתוך תוכניות ה-BASIC כך, שבסיום היחידה תוכל לפתח את המשחק "ירי בצלחות חרס" תוך שימוש בתוכנה "SPRITES" ששימשה אותך ביחידה 6.

**שים לב!** ביחידת לימוד זאת ננצל את שתי התוכנות "HIGH-RESOLUTION" ו-"SPRITES" של "מחשבת" אשר שימשו אותך כבר ביחידות הקודמות.

בברכת המשך לימוד

נעים ומרתק

דני קדם ויצחק קליסקי

"מחשבת" - מערכות למידה

# פרק א

## אותיות קידוש לבנה

### מבוא

בפרק זה נעסוק בהגדלה (ובהקטנה...) של אותיות וסימנים, כך, שתוכל לעצב אותיות בגדלים ובצורות כיד הדמיון הטובה עליך. הנה כמה דוגמאות:

123

123

1 2 3 123

### שים לב!

לצורך לימוד פרק זה עליך לטעון תחילה את התוכנה "HIGH RESOLUTION" של "מחשבת" (זו ששימשה אותך ביחידות 4 ו-5).

### PIXEL – פִּיקְסֵל

על מנת לעצב את כל האותיות היפות שלמעלה תזדקק לפקודה בוספת הנמצאת בתוכנה "HIGH RESOLUTION" של "מחשבת", שעדיין לא עשית בה שימוש, וזאת פקודת ה-PIXEL (פיקסל).

- ☐ עבור למסך הגרפי, נקה אותו, והדלק בו את הפיקסל  $10, 10$ .  
(תשובה 1)
- ☐ חזור למסך הטקסט (אינך זוכר כיצד? ראה חוברת 4 עמוד 7).

? האם כעת הפיקסל  $10, 10$  עדיין "דלוק"?  
תוכל לחזור למסך הגרפי ולבדוק את השערתך.

☐ כתוב (במסך הטקסט):

$P = \text{PIXEL}(10, 10)$

? מה ערכו של P כעת? (זכור הפיקסל  $10, 10$  - דלוק)  
☐ בדוק זאת על המחשב.

? מה ערכו של A אם תפקוד:

$A = \text{PIXEL}(20, 2)$

☐ בדוק על המחשב.

☐ כבה את הפיקסל  $10, 10$ .

(תשובה 2)

? מה ערכו של P כעת?  
☐ פקוד שנית:

$P = \text{PIXEL}(10, 10)$

? מה ערכו של P כעת?

### נסכם:

- הפקודה  $\text{PIXEL}(X, Y)$  מאפשרת לבדוק את מצבו של הפיקסל  $X, Y$ :
- אם הפיקסל  $(X, Y)$  דלוק - אז:  
 $\text{PIXEL}(X, Y) = \underline{\hspace{2cm}} (0, 1)$
- אם הפיקסל  $X, Y$  מכובה - אז:  
 $\text{PIXEL}(X, Y) = \underline{\hspace{2cm}} (0, 1)$

## 2 הערות חשובות

1. אי אפשר לכתוב:

PRINT PIXEL (X, Y)

חייבים להכניס תחילה את הערך של PIXEL (X, Y) לתוך משתנה כלשהו (נסה!)

2. הקיצור של PIXEL הוא, כרגיל P ו-SHIFT של I - אבל במקרה שכותבים בקיצור אין צורך (אסור) לכתוב את הסוגריים הראשונים, אלא יש לכתוב כך:

B = P  $\uparrow$  X, Y  
SHIFT I

כעת אתה יכול לחזור למשימה - כתיבת אותיות ענק - ולבצע אותה. יש בידך כל הכלים הדרושים לכך! מתקשה? נפתח את התוכנית שלב אחר שלב:

### שלב 1


כתוב תוכנית שתדפיס בפינה 

השמאלית-תחתונה של המסך

הגרפי את האות - A:

(העזר ברשת המסך הגרפי

המופיעה ביחידה 4 עמוד 74).




(תשובה 3)

כפי שידוע לך, כל אות או סימן מודפסים על המסך בתוך משבצת של 8x8, כלומר 64 פיקסלים. (ראה חוברת 4 עמוד 29)



בתוך משבצת זו המחשב מדליק רק את הפיקסלים היוצרים יחד את הסימן הדרוש:




- ☐ התבונן היטב באות A שעל המסך, (הדפס אותה גם בנגטיב כדי לראות היטב את המשבצת כולה).
- ☐ צייר בתשבץ שמשמאל את הפיקסלים היוצרים את האות A כפי שהיא מופיעה במסך:

(תשובה 4)

## שלב 2

בחזור לתוכנית שלנו. הוסף עוד קטע לתוכנית כך, שהמחשב יסרוק, פיקסל אחר פיקסל את כל 64 הפיקסלים של המשבצת בה מודפסת האות A. בכל פעם שהמחשב נתקל בפיקסל "דלוק" (העזר בפקודת (PIXEL), הוא מדליק משבצת שלמה במקום המתאים:



## שים לב!

יתכן שאתה מתקשה לבצע שלב זה. לפניך מספר רמזים:

- אתה זקוק לשורה בערך כזאת:

IF... THEN PRINT AT...


המחשב לא יקבל שורה כזאת. מדוע? ראה חוברת 5 עמוד 61, תחת הכותרת הערה חשובה.

- אתה עלול לקבל אות גדולה - אבל הפוכה. הדבר נובע מכך שהראשית של הפיקסלים היא למטה-משמאל, והראשית של המשבצות היא בפינה השמאלית-עליונה של המסך.

(ראה את רשת המסך הגרפי ביחידה 4 עמוד 74)

(תשובה 5)


## משכללים את התוכנית

הוסף לתוכנית שיכלול כך, שהמחשב יחכה ללחיצה ממך:   
כל סימן עליו תלחץ, המחשב מנקה את המסך הגרפי ומדפיס עליו,  
בגדול, את הסימן.

(תשובה 6)

### שכלול נוסף

במקום להדפיס משבצות מלאות, שנה את התוכנית כך, שהמחשב יבנה  
את האות הגדולה מאותיות קטנות, כך:



כל אות גדולה  
בנויה מהאותיות  
הקטנות שלה:  
• לחץ גם על מקשים  
של סימנים גרפיים.

(תשובה 7)

### שולטים על גודל האות

עד כה הגדלת כל סימן או אות פי 8 בלבד.  
כעת נשנה את התוכנית כך, שתוכל להגדיל כל סימן פי כמה שתרצה -  
בגובה או ברוחב.


תוכל אפילו להקטין את האותיות!

לשם כך נתאר את המשימה בשלבים - מן ההתחלה:

## שלב 1

1

כתוב תוכנית שתבצע את הפעולות הבאות:  
התוכנית מדפיסה את האות "A"  
במשבצת השמאלית-תחתונה של המסך:



2

התוכנית סורקת, פיקסל אחר פיקסל,  
את כל 64 הפיקסלים של המשבצת בה  
מודפסת האות "A", ובעזרת פקודת  
PIXEL מעתיקה את האות A למקום  
אחר במסך.  
(פקודת PLOT תבוא לעזרתך)



(תשובה 8)

## הערה חשובה

גם אם הצלחת לפתור את שלב 1, התבונן בתשובה 8.  
אתה רואה שם, נוסף למשתנים X ו-Y, עוד שני משתנים: A ו-B.  
יכולנו אמנם להשתמש ב-X וב-Y גם עבור קריאת האות שהודפסה וגם  
עבור ה-PLOT של האות שהועתקה, אך העדפנו להשתמש בשני משתנים  
בוספים, A ו-B. הדבר יאפשר לנו, כפי שתיווכח בהמשך, להגדיל את  
האות.

לכן - אם ביצעת את שלב 1 רק עם X ו-Y - שנה את התוכנית תוך  
שימוש בשני משתנים בוספים, כמו בתשובה שלנו.

## שלב 2

שנה את התוכנית כך, שהאות A המועתקת תתקבל ברוחב כפול:



מתקשה?

## רמז א

מימין לכל PLOT פקוד על המחשב לבצע עוד PLOT.

## רמז ב

אם רק ביצעת PLOT נוסף, ולא דאגת לשינוי נוסף בתוכנית, קיבלת, קרוב לודאי, אות A שמנה:



ולא אות A רחבה:



כדי לקבל A רחבה, כנדרש, עליך לגרום למשתנה A לגדול ב-2 בכל פעם.

(תשובה 9)

שחק עם התוכנית כדי לקבל אות רחבה פי 3, פי 4 ועוד. □

## ארגון התוכנית

מנסיונך אתה כבר יודע, כי אין דרך אחת לכתוב תוכנית. הנה גירסה בוספת לתוכנית שלך המגדילה את האות A:

```
10 A=0 : B=50
20 GRAPHICS : CLS
30 PRINT AT 0, 24; "A"
35 REM SCANNING THE ORIGINAL CHARACTER"
40 FOR Y=0 TO 7
50 FOR X=0 TO 7
60 IF PIXEL (X,Y)=1 THEN GOSUB 200
70 A=A+2
80 NEXT X
90 A=0 : B=B+1
100 NEXT Y
110 END
190 REM *****
200 REM "PRINTING THE BIG CHARACTER"
210 PLOT A, B
220 PLOT A+1, B
250 RETURN
260 REM *****
```

תוכנית זאת מבצעת בדיוק אותה פעולה כמו התוכנית שהובאה בתשובה 9.

במה נבדלת תוכנית זאת מהקודמת?

ההבדל העיקרי הוא בארגון התוכנית. אנו מתארים לעצמנו כי כבר ניתקלת בבעיה של יכולת קריאת תוכנית מחשב. אם התוכנית אינה מאורגנת היטב, ישנם קשיים בקריאת התוכנית אם רוצים לאתר בה בגים ולהכניס שינויים. הסיבה לכך שאנחנו ארגנו את התוכנית מחדש נעוצה בעובדה, כי ככל שנשפר ונשכלל את התוכנית, נצטרך לשתול עוד ועוד פקודות בשורה 60 - דבר שיהפוך את קריאת התוכנית ועריכתה למייגעת ומסורבלת.

## כמה הערות לגבי התוכנית המאורגנת

- שים לב לשורות בהם מופיעה הוראת ה-REM. אתה כבר יודע, כי כאשר המחשב מגיע לשורה עם הוראת REM הוא מדלג אוטומטית לשורה הבאה: (ראה יחידה 3 עמוד 32) השורות עם הוראת REM באות רק כדי לעזור לנו לקרוא ולהבין מה מבצעות השורות השונות בתוכנית.
- משורה 35 ועד שורה 100 המחשב סורק את הפיקסלים הבונים את האות המקורית שהודפסה בפינה השמאלית התחתונה של המסך.
- משורה 200 ואילך התוכנית מדפיסה את האות המשוכפלת בהתאם לגודל הרצוי. שים לב, כי המחשב פונה לבצע את שורות 200 ואילך רק כאשר התבאי שבשורה 60 מתקיים.

- משורה 100 ואילך התוכנית מדפיסה את האות המשוכפלת בהתאם לגודל הרצוי.
- שים לב, כי המחשב פונה לבצע את שורות 100 ואילך רק כאשר התנאי שבשורה 40 מתקיים.

- ☐ שנה את התוכנית שלך לפי התוכנית החדשה ובדוק אם היא אכן משרטטת את האות המוגדלת.
- בשורה 110 מופיעה ההוראה END (הוראה זאת נלמדה ביחידה 3, עמוד 29).
- ☐ מדוע הוכנסה שורה זאת לתוכנית?
- ☐ נסה להריץ את התוכנית ללא שורה 110.

### שלב 3

- ☐ שנה את התוכנית כך, שלפני ציור האות המחשב ישאל אותך איזו אות אתה רוצה להרחיב, ופי כמה אתה רוצה להרחיב אותה.
- (תשובה 10)

- קרא למשתנה ההרחבה - MX.
- בדוק את פעולת התוכנית:
- הכנס הרחבה פי 1
  - הכנס  $MX = 0.8$
  - הכנס  $MX = 2$  ...

### שלב 4

- כדי להגדיל את מהירות הציור של האות, החלף את פקודות ה-PLOT שבתוך לולאת ה-FOR-NEXT בפקודת DRAW פשוטה.
- ☐ הרץ את התוכנית ובדוק את התוצאות.
- (תשובה 11)
- ☐ איזה שינוי עליך להכניס לתוכנית כך שעבור  $MX=1$  המחשב ישכפל את האות בדיוק בגודלה המקורי ללא שינוי.
- (תשובה 12)

### שלב 5

- כעת הגיע הזמן להגדיל את האות גם כלפי מעלה.
- נסה לעשות זאת בכוחות עצמך, לפני שתפנה אל הרמזים.

## רמז א

הוסף, מיד לאחר ה-PLOT וה-DRAW, עוד פקודות PLOT ו-DRAW כך, שיצטייר קו נוסף מעל לקו הקודם.

## רמז ב

זוכר שקיבלת אות עבה במקום אות רחבה? המנע מבג דומה ובצע שינוי מתאים בשורה שמגדילה את המשתנה B.

(תשובה 13)

## שלב 6

שנה את התוכנית כך, שהמחשב ישאל אותך בהתחלה 3 שאלות:

- איזו אות להגדיל?
  - פי כמה להגדיל לרוחב (MX)?
  - פי כמה להגדיל בגובה (MY)?
- וכמובן יבצע את ההגדלות כפי שפקדת - ויחזור לשאול אותך על אות חדשה.

(תשובה 14)

## שחק עם התוכנית

- הדפס אותיות בגדלים שונים.
- הקטן אותיות.
- הגדל סימנים גרפיים.
- שנה את  $\theta$ , DRAW MX ל-3, DRAW MX וראה מה מתקבל.

## אתגר

נסה לקבל אות כזאת:

כלומר עשויה מפסים ולא מלאה.



(תשובה 15)

## התוכנית מגדילה משפטים שלמים

כעת נשכלל את התוכנית כך שנוכל להגדיל לא רק אות אחת, אלא מילה שלמה או כמה מילים.  
לשם כך עליך להרחיב את ידיעותיך על מחרוזות:

### MID\$ – מידדולר

כתוב במחשב: ☐

`X$ = "ABCDEF"`

אח"כ כתוב: ☐

`PRINT MID$ (X$,3,2)`

בתגובה המחשב כתב \_\_\_\_\_.

מה הבין המחשב בפקודה `PRINT MID$ (X$, 3, 2)` ?

האות ה-3 ↓ "A B C D E F" └─┬─┘ 2 אותיות	המחשב הבין שהוא צריך להדפיס מתוך המחרוזת X\$, החל מהאות ה-3 שתי אותיות:
---	---

פקוד על המחשב: ☐

- להדפיס את האותיות BCD מתוך המחרוזת X\$.
- להדפיס את האות החמישית מתוך X\$.

(תשובה 41 בעמוד 63)

### חידה

מה יקרה אם תשמיט את המספר האחרון שבסוגריים: ?

`PRINT MID$ (X$, 3)`

נסה! ☐

אתה רואה שללא המספר השני בפקודת ה-MID\$, המחשב מדפיס את קטע המחרוזת מן המספר הראשון עד הסוף.



## משימה

כתוב תוכנית קצרה (משורה 300 ואילך, למשל) שתבצע את המשימה ☐

הבאה:

1. המחשב שואל אותך (INPUT) למחרוזת.
2. המחשב מדפיס אותה על המסך, אבל - מהסוף להתחלה!

למשל:

- אתה מכניס "A3BC"
- המחשב מדפיס CB3A
- (תוכל להעזר ב-LEN. ראה יחידה 2 עמוד 42)

(תשובה 17)

## נסכם

- המספר הראשון בפקודת MID\$ אומר \_\_\_\_\_ לקרוא במחרוזת.  
(החל מאיזה סימן, כמה סימנים)
- המספר השני בפקודת MID\$ אומר \_\_\_\_\_ לקרוא במחרוזת.  
(החל מאיזה סימן, כמה סימנים)

**הערה** הקיצור של MID\$ הוא M ו-I SHIFT: אין צורך לכתוב את הסימן \$ אך יש צורך בסוגריים.

## אפשר לחזור לתוכנית

שבה את התוכנית המגדילה אותיות כך, שתוכל להכניס מילים שלמות ☐  
והתוכנית תגדיל את כל האותיות, כך:

12345

אם אתה מתקשה בביצוע המשימה אנו מציעים כי תפנה עכשיו לביצוע משימת הביניים הבאה:

## משימת ביניים

שנה את התוכנית כך ש:

- לאחר שהמחשב יקלוט מחרוזת עם מספר אותיות, הוא ידפיס את האותיות הגדולות בזו אחר זו - באותו מקום על המסך. (זאת, כמובן, לאחר מחיקת האות הקודמת.) (תשובה 18)


סיימת בהצלחה את משימת הביניים?

עכשיו עליך לשנות את התוכנית כך שהמחשב ידפיס את האותיות אחת ליד השניה.

עדיין זקוק לרמז?

- עליך לזכור כי המחשב סורק את כל הפיקסלים הבונים את האות המקורית (8×8).
- כל פיקסל מוגדל לרוחב פי MX.
- כדי שהאות הבאה תודפס מימין לאות שהודפסה קודם, המשתנה A צריך להשתנות ב: \_\_\_\_\_

(תשובה 19)

ועכשיו השלם את התוכנית. 

(תשובה 20)

## רווחים בין האותיות

שמת לב בודאי, כי כשהאותיות גדולות מאוד יש רווח גדול בין האותיות.

שפר את התוכנית כך, שתוכל לשלוט על הרווח (מספר הפיקסלים) שבין אות לאות.

כדי לבדוק אם התוכנית פועלת כשורה:

פקוד על המחשב להדפיס את המחרוזת "123" בהגדלה של פי 8 (לאורך ולגובה) כך שיתקבל:

123

(תשובה 21)

## משימה

כתוב את שמך ושם משפחתך על המסך - גדול ככל האפשר!

## אתגר

שנה את התוכנית כך:

1. אתה מכניס את שמך.
2. המחשב מחשב, לפי אורך השם, פי כמה להגדיל אותו כך שימלא את כל רוחב המסך.
3.  $R=\emptyset$ ,  $MY=MX$  (תשובה 22).

## עיצוב אותיות

כהרגלנו נתחיל לשחק עם התוכנית על-מנת לעצב אותיות מצורות שונות ומשונות...


שינוי קל בתוכנית יגרום להדפסת אותיות כאלו:

### משימה 1



(תשובה 23)

### משימה 2



### משימה 3

האם תוכל לשנות את התוכנית כך, שתדפיס אותיות וסימנים כאלה:



(תשובה 24)

- ללא תלות בגודל האות - כל פס אופקי באות מורכב משני קווים:
- הזכר באותיות שקיבלת עמוד 15 שהיו מורכבות מפסים.

## אתגרים לסיום

### • שעון דיגיטלי גדול

זוכר את תוכנית השעון הדיגיטלי מחוברת "מחשבת-2"? כעת, כאשר אתה יודע לצייר ספרות גדולות, תוכל לבנות שעון דיגיטלי המראה את השעה המדויקת בספרות ענקיות!

### משחקים בינגו!

עכשיו, כאשר אתה יכול להגדיל אותיות ומספרים, המחשב יכול לעזור לך בניהול משחק בינגו:

- המחשב מגריל את המספרים המשתתפים במשחק הבינגו שלך. ומציג אותם בגדול על המסך.
- כאשר יש "בינגו" (כלומר שורה, או עמודה או לוח שלם) תוכל לבדוק בעזרת המחשב אם אכן זהו בינגו אמיתי.
- לאחר שהמחשב מצא שהמספרים נכונים מופיעה על המסך הכותרת: "בינגו" (BINGO)
- אם תהיה מעוניין, המחשב ישאל לשמו של הזוכה, ולאחר שתכניס את שמו המחשב ידפיס אותו באותיות גדולות על המסך.
- וכמובן תוכל להכניס עוד ועוד שכלולים כרצונך.


# פרק ב

## משחקי מספרים

### משחקי מספרים

#### תאור המשחק

אתה עומד לפתח תוכנית שתבצע את המשחק הבא:



המחשב מגריל 4 ספרות באקראי

(בדוגמא שלנו - 4,2,6,2)

ומציג אותן על המסך.

מספר נוסף מופיע מתחתן

(בדוגמא שלנו - 20).

עליך לנחש ולחשב אילו פעולות חשבון יש להציב בין הספרות כך,

שתוצאת התרגיל החשבוני תשווה למספר הרשום על המסך (במקרה


שלנו - 20).

(תרגילים מסוג זה ביצעת בחוברת 4, עמוד 53).

שים לב! אתה יכול להשתמש ב-4 פעולות החשבון: חיבור, חיסור

כפל וחילוק יותר מפעם אחת ואינך חייב, כמובן, להשתמש בכלן.

האם אתה יודע לפתור את הדוגמא שלמעלה?



לאחר שניסית לפתור את התרגיל -

אתה לוחץ על מקש כלשהו והמחשב

מציג לך על המסך תבנית אפשרית

של תרגיל חשבון שייתן את הפתרון

הנכון.

במקרה שלנו המחשב הדפיס את התבנית:  $S = A*B+C*D$

ואכן, אם תציג את המספרים במקומות המתאימים תקבל 20:

$$4*2+6*2 = 20$$

הערה: האם אתה מבין מדוע תרגיל חשבוני זה אכן נותן 20 ולא

28? אם אינך יודע, אנו ממליצים כי תפנה ללמוד את

הנספח: "סדר ביצוע פעולות חשבון", שבחוברת 4.

## שני דרכים לאותה מטרה

איך המחשב פותר את התרגיל? אפשר לחשוב על 2 דרכים אפשריות:

### דרך א

המחשב מגריל את 4 הספרות ומגריל גם את התוצאה. כעת הוא מנסה (כמוך) להציב פעולות חשבון בין הספרות על-מנת להגיע לתוצאה שעלתה בגורל.

### דרך ב

המחשב מגריל את 4 הספרות וכן את פעולות החשבון שביניהן. בהתאם לכך הוא מחשב את תוצאת התרגיל ומציג אותה על המסך. כמוכן שבשיטה זאת המחשב יודע מראש את הפתרון ... (את פעולות החשבון).  
אנו נבחר בדרך השנייה, מכיוון שהיא קלה יותר לתיכנות ומהירה יותר - וגם מבטיחה שיש פתרון.

### מה אתה כבר יודע ומה עליך ללמוד

- אתה כבר יודע איך לפקוד על המחשב להגריל 4 ספרות באקראי, ולהציג אותן על המסך.
- אינך יודע איך לפקוד על המחשב להגריל פעולות חשבון באקראי. לשם כך עליך ללמוד את הפקודה:

### ON... GO TO...

הכנס את התוכנית הבאה למחשב והרץ אותה:



```
10 PRINT "{CLR/HOME}"
20 INPUT R
30 ON R GO TO 40, 50, 60, 70
40 PRINT "AAA" : GO TO 20
50 PRINT "BBB" : GO TO 20
60 PRINT "CCC" : GO TO 20
70 PRINT "DDD" : GO TO 20
```

כשהתוכנית רצה ונמצאת ב-INPUT הכנס באקראי את אחד המספרים: 1 עד 4 וראה כיצד מגיב המחשב.



## הסבר

השורה החדשה היא, כמובן, שורה 30:

```
30 ON R GO TO 40, 50, 60, 70
```

והיא גורמת למחשב לפנות לשורה מתאימה בתוכנית בהתאם לערך של

R. שורה זאת מחליפה, למעשה, בבת אחת 4 שורות של IF:

```
IF R=1 THEN GO TO 40
```

```
IF R=2 THEN GO TO 50
```

```
IF R=3 THEN GO TO 60
```


```
IF R=4 THEN GO TO 70
```


## משימה

אם הבנת את ההסבר - בצע שינוי קטן בשורה 30 כך, שכאשר תלחץ


2 המחשב ידפיס "CCC", וכאשר תלחץ 3 המחשב ידפיס "BBB".

(תשובה 25)

מה יקרה אם תלחץ על הסיפרה 5 או 9? 

נסה. 

לכאורה המחשב מגיב כאילו לחצת על 1. אבל:

הוסף לתוכנית את השורה הבאה: 

```
35 PRINT "***" : GO TO 20
```

הרץ את התוכנית. לחץ על הספרות 1-4. לחץ גם על ספרות אחרות. 

אתה רואה כי, כאשר אתה לוחץ על סיפרה הגדולה מ-4 המחשב מתעלם

משורה 30 וממשיך הלאה לשורה 35.

## מסקנה

אם בפקודת ... ON X GO TO כתבת 6 מספרי שורות אפשריות (אחרי

ה-GO TO) והמשתנה X קיבל את הערך 7 או יותר - המחשב מתעלם

מכל השורה!

שנה את התוכנית כך, שהמחשב לא יחכה שנכניס לו ב-INPUT את המספר, אלא יגריל באקראי מספר בין 1 לבין 4, ובהתאם לתוצאות ההגרלה יפנה לשורה מתאימה בתוכנית.

(תשובה 26)

## חזרים אל משחקי המספרים...

כעת יש בידך כל הכלים הדרושים כדי לכתוב את תוכנית המשחק, כפי שהוצגה בעמוד 21.

נסה לכתוב את התוכנית - לפני שתפנה אל הרמזים הכתובים בהמשך.

### שלב 1

כדי לבצע את שלב 1 מעמוד 21 על המחשב לבצע את הפעולות הבאות:

- להגריל ערך אקראי (מ-1 עד 9) לכל אחד מהמשתנים C, B, A ו-D.
- לפנות באקראי (ON... GO TO...) אל אחת מן השורות:

110 S = A\*B+C-D : PRINT S : GO TO 180

120 S = A+B+C-D : PRINT S : GO TO 180

130 S = A/B+C\*D : PRINT S : GO TO 180

אתה יכול, כמובן, להמשיך ולהוסיף שורות כאלה כדי לקבל יותר ויותר צירופים אפשריים של פעולות חשבון.

(תשובה 27)

### שלב 2

כעת עליך להוסיף עוד קטע לתוכנית, שיבצע את שלב 2 של המשחק (ראה עמוד 21):

בלחיצה על מקש כלשהו המחשב פונה אל שורה מתאימה מתוך קבוצת שורות, שהן העתק כמעט מדויק של תכניות התרגילים שבשורות הקודמות:

210 PRINT "S=A\*B+C-D" : GO TO 300

220 PRINT "S=A+B+C-D" : GO TO 300

230 PRINT "S=A/B+C\*D" : GO TO 300



## שים לב לנקודות הבאות

- סדר הפעולות בשורות 210 ואילך חייב להיות זהה לסדר הפעולות בקבוצת השורות 110 ואילך, על מנת שהתשובה תהיה נכונה.
- אם כתבת שורה כזאת:

ON R GO TO 110, 120, 130

אז אותו R ישמש אותך בשורה:

ON R GO TO 210, 220, 230

(תשובה 28)

## אגב

אם המחשב הציג בפניך את הבעיה, ואתה רוצה להשתמש במחשב עצמו לצורך חישובים:

- עצור את ריצת התוכנית
- בצע את החישובים הדרושים (PRINT 4/2+6\*2)
- כדי להמשיך מהמקום שהמחשב עצר, פקוד:

(ENTER +) CONT

שפירושו "המשך בביצוע התוכנית מן המקום בו הפסקת".  
(זהו קיצור של המילה האנגלית CONTINUE - המשך)  
ואז המחשב ימשיך בריצת התוכנית ותוכל לראות את תשובתו.

## משכללים את התוכנית

בודאי הפריעה לך העובדה שחלק מן התוצאות היו שברים-עשרוניים.  
הוסף עוד שורה שתגרום למחשב לבדוק את תוצאת התרגיל.  
אם התוצאה אינה מספר שלם - החזר אותו להגרלה!

## רמז

❓ כיצד תבדוק אם משתנה כלשהו הוא מספר שלם או לא?  
פקודה שלמדת בחוברת 3, עמוד 43 תבוא לעזרתך.  
(תשובה 29)



ואולי תרצה שהתוצאה תהיה מספר חיובי דווקא?  
לשם כך הוסף שורה כזאת, שלא תאפשר למחשב להציג מספרים שליליים.  
(תשובה 30)

## **האמת היא, שהתוכנית די מוגבלת...**

ודאי שמת לב כי התוכנית די מוגבלת - היא מנסה רק את מספר  
הצירופים שכתבת עבורה!  
תוכנית יותר מתוחכמת היא זאת אשר תגדיל באקראי הן את המספרים  
והן את הפעולות - וכך נקבל את כל מספר הצירופים האפשרי.  
לשם כך עליך ללמוד על "מחסנית המקשים".

פרק ד' יוקדש כולו ללמוד מחסנית המקשים וכך תוכל בין השאר  
לחזור ולשכלל את תוכנית משחקי המספרים.

# פרק ג

## התוכנית הגדולה – משחקי מילים

ב"משחקי-מילים" אנחנו מתכוונים לאוסף תוכניות, אשר בעזרתן תוכל לתרגל, בדרך משעשעת, כתיבה נכונה של מילים אנגליות ועבריות. אנחנו מתארים לעצמנו, כי כבר עכשיו יש לך מספר רעיונות למשחקי-מילים. הנה כמה הצעות שלנו למשחקי-מילים.

1. **הבוקי מילים:** מילה מופיעה למשך זמן קצר ובעלמת – עליך לחזור ולכתוב את המילה.

2. **השלמת אותיות:** מילה מופיעה על המסך כאשר מספר אותיות חסרות כמו: C \_ \_ AR עליך לנחש מהן האותיות החסרות.

3. **נחש את המילה:** על המסך מופיעים קווים כמספר האותיות. כמו למשל: \_ \_ \_ \_ \_ עליך לנחש את המילה.

4. **אותיות מבובלות:** המחשב מציג את האותיות בצורה מבובלת (למשל: EFTHAR) על המסך, ועליך לנחש את המילה הנכונה (FATHER).

5. **תרגום מילים:** מופיעה מילה בעברית ועליך לכתוב אותה באנגלית (כמובן שאפשר בצורה הפוכה – מאנגלית לעברית). אם המחשב שלך אינו יודע לכתוב עברית לא נורא! ביחידת הלימוד הבאה תלמד כיצד להפוך את הקומודור למכונת כתיבה עברית. הבאנו כאן רק כמה רעיונות לסוגים שונים של "משחקי-מילים". האם כבר בשלב זה יש לך רעיונות נוספים ל"משחקי-מילים"?

# משחק 1 – הבזקי מילים

נתחיל בפיתוח המשחק הראשון...

נפתח את הפרק בפיתוח משחק קצר הנקרא "הבזקי מילים".

1

אתה מלמד את המחשב מספר מילים.

לצורך הדגמה נבחר מספר מילים השגורות בתחום המחשבים:

RESOLUTION, INTERFACE, MICROPROCESSOR, SUBROUTINES, INPUT,  
DATA, READ, MEMORY, PERIPHERIAL, REGISTER, POINTER,  
HARDWARE, SOFTWARE.

המחשב בוחר באקראי את אחת המילים שלימדת אותו ו"מבזיק" אותה על המסך לזמן קצר.

2

עליך לזהות את המילה ולכתוב אותה נכון:

- אם צדקת - המחשב עובר למילה הבאה.
- אם טעית - המחשב מודיע לך כי טעית - ומבזיק אותה שנית, לפרק זמן ארוך יותר.

לדעתנו אתה יכול לגשת לבנות את המשחק הזה, ללא צורך ברמזים שלנו. נסה!

ואם בכל זאת אתה מתקשה...

## שמוש במערכים מחרוזתיים

מאחר שבמשחק אנו מעוניינים להבזיק מילים באופן אקראי, רצוי מאוד לטעון את המילים לתוך מערך - ובאופן כזה ניתן להציג אותן באופן אקראי על המסך.

כמו שיש מערכים מספריים (אינך זוכר? ראה יחידה 3 עמוד 49) יש גם מערכים מחרוזתיים!

למשל: DIM A\$(3)

A\$(1) = "COMMODORE"

A\$(2) = "COMPUTER"

- גם במערכים מחרוזתיים אין צורך להגדיר (DIM) מערך מחרוזתי של עד 10 מחרוזות.
- אורך כל מחרוזת - מכסימום 256 סימנים.



האם עכשיו אתה יכול לגשת ולפתח את המשחק? - בסה!

אם לא - נפתח את המשחק צעד אחר צעד:

## קליטת מילים

כאמור, בקטע הראשון של המשחק עליך "ללמד" את המחשב את המילים, שאתה רוצה שישתתפו במשחק.

ובמילים אחרות: עליך לטעון מערך מחרוזתי במילים כרצונך.

כיצד תעשה זאת?

ניתן לחשוב על מספר דרכים כדי לטעון את המערך המחרוזתי במילים.

לפניך קטע של תוכנית, שבעזרתו תוכל לטעון את המערך.

בתוכנית חסרים כמה פרטים. השלם אותם:

1Ø DIM W\$(3Ø)

2Ø FOR \_\_\_\_\_

3Ø INPUT \_\_\_\_\_

4Ø NEXT

השלם את החלקים החסרים.



(תשובה 31)

להצעה זאת יש נקודת תורפה רצינית.

- בכל פעם שתריץ את התוכנית תצטרך לטעון מחדש את המערך - וזה תהליך די מיגע.

כיצד תתגבר על כך?

אחד הפתרונות הוא להשתמש ב-READ-DATA כדי לטעון מערך מחרוזתי במילים.

(כזכור READ-DATA בלמד ביחידה 3 עמוד 35)

בכל פעם שתריץ את התוכנית, המחשב יטען מחדש את המערך המחרוזתי

מתוך שורת-הנתונים (DATA) - וכך לא תצטרך לדאוג פן המחשב

"ישכח" את המילים החדשות!

## והערה אחרונה

ברור כי בכל משחקי המילים בצטרך להגדיל מילים.  
לשם כך בצטרך לדעת את מספר המילים המשתתפות בהגדלה.  
כתוב את קטע התוכנית הקולטת את המילים באופן כזה, שהמחשב יספור  
בעצמו את מספר המילים המשתתפות במשחק כדי שהוא יוכל להגדיל  
ביניהן.

(תשובה 32)

ועכשיו השלם את תוכנית המשחק – "הבזקי מילים".  
**הערה** רצוי שמשחק 1 יתחיל בשורה 200. בהמשך תבין מדוע.  
(תשובה 33)

ברור כי בשלב זה יש לך כבר כמה הצעות לשיפור המשחק "הבזקי-  
מילים". העזר בסבלנות – בהמשך נחזור למשחק, לשיפורים.

## התוכנית הגדולה

לפני שניגש לתכנת את המשחק השני, נעצור לרגע ונשאל את עצמנו  
כיצד היינו רוצים לראות את התוכנית הסופית?  
• האם היינו רוצים להריץ כל משחק מילים לחוד?  
• האם היינו רוצים שכל "משחקי-המילים" יופיעו בתוכנית אחת  
גדולה?

אנו בחר בדרך השניה – נצרף את "משחקי-המילים" שנפתח לתוכנית  
אחת גדולה! מה יהיה המבנה הכללי של תוכנית זאת?  
החלקים העיקריים בתוכנית יהיו:

- **קליטת מילים** בחלק זה של התוכנית יקלוט המחשב את המילים  
אותן נהיה מעוניינים לתרגל ב"משחקי-המילים". (אתה כבר תכנתת  
חלק זה)

● **תפריט (MENU):** על המסך תופיע רשימה של משחקי-המילים.


לאחר שתבחר במשחק בו אתה מעוניין - המחשב יפנה לתוכנית המתאימה...

● **משחקי מילים:** חלק זה הוא אוסף של התוכניות השונות של "משחקי-המילים".

בכל שלב של המשחק תוכל לפקוד על המחשב לחזור ל"תפריט" ותוכל לבחור במשחק-מילים אחר.

## מסך התפריט

את קטע התוכנית אשר קולטת את המילים המשתתפות כבר פיתחת. עכשיו עליך לכתוב את קטע התוכנית המציג את "התפריט": לאחר שהמילים יקלטו לתוך מערך מחרוזתי המחשב מציג את רשימת המשחקים:



לעת עתה התפריט שלך די דל. יש ברשותך משחק אחד בלבד. בהמשך, כאשר תפתח עוד משחקי מילים, התפריט יהיה יותר "עשיר".

- כתוב את מסך התפריט כך, שבלחיצה על המספר 1 המחשב יפנה למשחק 1.

לחיצה על 2 תפנה אותך למשחק 2 (שעדיין לא קיים) - וכך הלאה.


מתקשה?

פקודות GET ו-ON GO TO יבואו לעזרתך.

(תשובה 34)

עכשיו נמשיך בפיתוח משחקים נוספים.

## משחק 2 – אותיות חסרות



1

המחשב מציג על המסך את אחת המילים  
שלימדת אותו – אבל אות אחת, אקראית  
חסרה ובמקומה יש מקף:

2

עליך לכתוב את המילה הנכונה  
בשלמותה והמחשב יודיע לך אם  
כתבת אותה נכון או לא:

לשם כך עליך לדעת איך אפשר להתיחס לכל אות או סימן בתוך מחרוזת  
(העזר ב-MID\$ עליו למדת בפרק א עמוד 16).

- שים לב להתחיל את המשחק בשורה מתאימה כך, שהמחשב אכן יפנה  
ממסך התפריט למשחק 2 לאחר לחיצה על הסיפרה 2.  
(תשובה 35)

עוברים ממשחק למשחק דרך מסך התפריט...

בניח כי אתה במצא במשחק 1 ומעוניין לעבור ולשחק במשחק 2. הכנס  
לתוכנית שינוי כזה שיאפשר לך לעשות זאת מבלי להריץ את התוכנית  
מחדש.

זקוק לעזרה?

הוסף שורה למשחק 1 כך, שבלחיצה על הסימן "<" (RETURN +) תחזור  
למסך התפריט. ומשם כמובן שאין בעיה להגיע למשחק 2.

(תשובה 36)

הוסף שורה מתאימה גם למשחק 2 שתאפשר לך לחזור למסך התפריט...






## משחק 3 – ניחוש מילים


1

המחשב מגריל מילה מאוצר המילים שלמד ומציג על המסך קוים (סימני "-") כמספר האותיות שבמילה: (נניח שהמילה NINE עלתה בגורל)




2

עליך לנחש מהי המילה. נניח שאתה אינך יודע מה המילה, ורוצה לבדוק אם האות "I" נמצאת בה. אתה לוחץ על המקש "I". מיד המחשב מדפיס את האות I במקומה הנכון:




3

אם אתה לוחץ על מקש של אות שאיננה במילה - המחשב אינו מגיב. אם אתה לוחץ על מקש של אות הנמצאת פעמים במילה - המחשב מדפיס אותה פעמים. נניח שלחצת על "N":



4

ברגע שכל המילה כתובה על המסך - אתה לוחץ על "הדק" (נניח הסימן "=") והמחשב עובר למילה הבאה:



ושוב - סיפקנו לך כבר את כל הכלים התיכנותיים הדרושים כדי לכתוב את התוכנית של משחק זה.

נסה לכתוב אותה בעצמך לפני שתפנה לתשובה...

(תשובה 37)

## קצת קוסמטיקה

עד כה, כאשר אתה עובר ממסך התפריט לאחד משלושת המשחקים, המחשב מתחיל מיד במשחק. רצוי שלפני כל משחק יופיע מסך פתיחה, המציין את שמו של המשחק...

## משכללים את התוכנית

לפני שתמשיך לפתח משחקי מילים נוספים, נעצור ונכניס כמה שכלולים למשחקים שכבר פיתחת.

נפנה תחילה לשכלול המשחק הראשון.

### שכלול המשחק הראשון

#### • דרגות קושי

ככל שמשך הזמן של המילים המובזקות על המסך קצר יותר, המשחק נעשה קשה יותר.

הכנס שכלול כזה, שיאפשר לך לבחור את משחק הבזקי המילים בדרגות קושי שונות.

(תשובה 38)

• ענית בפעם הראשונה בצורה נכונה – המחשב לא יציג מילה זאת שנית

#### אתגר

סביר להניח, כי ככל שמספר המילים הולך ומצטמצם, המחשב זקוק לזמן ממושך יותר בכדי להגדיל מילה, כי בהגרלה משתתפות, בינתיים, גם המילים שענית עליהן נכון בפעם הראשונה.

אם מילה כזאת עולה בגורל – המחשב צריך להתחיל את ההגרלה מחדש! שכלל את התוכנית כך, שהמילים שיוצאות מן המשחק לא ישתתפו כלל בהגרלה.

• סיימת לענות בצורה נכונה על כל המילים המובזקות?

המחשב יחזיר אותך לתפריט המשחקים הראשי.

• סיימת לענות בהצלחה על כל המילים?

המחשב ישמיע גם תרועת נצחון.

• ענית בצורה שגויה?

המחשב ישמיע צליל המזהה כי נעשתה שגיאה.

## שכלול המשחק השני

- דרגות קושי:  
הכנס שכלול כזה כך, שהמחשב יאפשר לך לבחור 3 דרגות קושי של המשחק: אות אחת, שתיים או שלוש חסרות.
- הכנס למשחק זה עוד שכלולים מאלה שהצענו לך במשחק הראשון.

## לסיום

בפרק זה התחלנו בפיתוח תוכנית גדולה, המורכבת ממספר רב של תת-תוכניות. לא סיימנו את פיתוחם של כל משחקי המילים (ראה עוד הצעות שלנו המופיעות בעמוד 27).

אנו משאירים לך להמשיך ולפתח את התוכנית, להכניס שיפורים בוספים וכמובן, להוסיף משחקי מילים משלך.

תוכנית זאת עשויה להיות שימושית מאוד כאשר תרצה לתרגל עצמך בכתיבה נכונה של מילים חדשות, אותן אתה לומד.

# פרק ד

## מחסנית המקשים

### מבוא

זוכר את המשחק "משחקי מספרים" שפיתחת בפרק ב? כבר אז אמרנו כי התוכנית די מוגבלת, מהסיבה שהיא מציגה רק את מספר הצירופים שכתבת עבורה. כעת נפתח תוכנית, שתגדיל את המספרים וגם את הפעולות, וכך נקבל משחק שבו מתקבלים כל הצירופים האפשריים של המספרים ופעולות החשבון! לשם כך עליך להשתלט על:

### מחסנית המקשים

כתוב את תוכנית-הלולאה הבאה במחשב:

```
10 FOR T=1 TO 8000 : NEXT
```

הרץ אותה - ובעוד היא רצה (ולפני שהופיעה הודעת READY), הקש במהירות על שלושת מקשי המספרים 9, 8 ו-7, כאמור לפני שהמחשב סיים להריץ את "התוכנית".

מה קורה כאשר המחשב מסיים לרוץ בתוך תוכנית הלולאה?

(המתן בסבלנות עד שהודעת ה-READY תופיע)

אם פעלת היטב, הרי שלאחר שהמחשב סיים את ריצת התוכנית והודיע READY, הוא המשיך והדפיס בבת אחת על המסך את 3 המקשים עליהם לחצת בזמן שהוא היה עסוק בהרצת התוכנית. כיצד קרה הדבר, שהמחשב "זכר" את המקשים עליהם לחצת בזמן שהוא רץ בתוך התוכנית?


לשם כך עליך להבין את סוד פעולתה של מחסנית המקשים.

ביחידה 3 פרק א' למדת, כי זכרוננו של המחשב בנוי ממספר רב של **תאי זכרון**. כל תא ממוספר. מספר זה הוא הכתובת הקבועה של תא הזכרון.

כל תא זכרון כזה נקרא בשפת המקצוענים בשם **BYTE – בַּיִת** אתה כבר יודע כי הבתים החל מ-1024 ועד 2024 אחראים לשמירת האינפורמציה שתודפס על המסך.

דע לך כי בזכרוננו של המחשב ישנם 10 בתים (תאי זכרון) המהווים את מחסנית המקשים. כתובתו של הבית הראשון של מחסנית המקשים הוא 631, והעשירי הוא 640.

כל מקש עליו לחצת, גרם להכנסת סימן מתאים למחסנית לפי סדר ההקשה:



המחשב "מסכים" להכניס את הסימנים עליהם לחצת, גם אם הוא רץ בתוך תוכנית.

כאשר המחשב סיים לרוץ בתוכנית, הוא "התפנה", ניגש למחסנית ושלף משם את הסימנים לפי סדר הופעתם במחסנית – והדפיס אותם על המסך.

מה המספר המכסימלי של סימנים שתוכל להכניס למחסנית המקשים תוך כדי הרצת התוכנית? ☐

בדוק השערתך במחשב. ☐

(תשובה 39)

וכששום תוכנית אינה רצה, והמסמן מהבהב – מדוע אינך מבחין בפעולתה של המחסנית?

הסיבה לכך היא, שהמחשב "פנוי" ושולף במהירות מן המחסנית את הסימן שהכנסת לתוכה ומדפיס אותו על המסך!

## **נסכם:**

- בזכרון המחשב יש "מחסנית מקשים", הבנויה מ-10 בתים. לחיצה על מקש כלשהו מכניסה את הסימן שלו לתא מתאים במחסנית המקשים.
- מיד כשהמחשב מתפנה מהרצת התוכנית, הוא מדפיס על המסך את הסימנים שבמחסנית.
- המחסנית יכולה להכיל עד 10 סימנים.

## **מכניסים סימנים למחסנית באמצעות פקודת POKE**

הסימנים נכנסו עד כה ל"מחסנית-המקשים" באופן אוטומטי, לאחר לחיצה על מקש כלשהו בלוח המקשים.

## **משימה**

עליך להכניס לתא הראשון של המחסנית את האות "A" כך שהיא תודפס על המסך מבלי ללחוץ על מקש A. לשם כך עליך להשתמש בטבלה הנמצאת בספר היצרן שקיבלת עם המחשב, בנספח F (APPENDIX F), הנמצא בעמוד 135. (יכול להיות שאצלך הוא במצא בעמוד אחר) בספח זה נקרא ASCII AND CHR\$ CODES. מה מספרה של האות A בטבלה זו?

(תשובה 40)

## **שים לב!**

יש שתי טבלאות עם מספרים של הסימנים. טבלה אחת כבר הכרת ביחידה ובה רשומים המספרים הסימניים של הסימנים עבור כתיבה במסך בעזרת POKE. (הטבלה נמצאת בספר היצרן תחת השם: SCREEN DISPLAY CODES)

יש סימנים שיש להם מספר שונה בכל טבלה. לכן, הקפד שלא להתבלבל בין שתי הטבלאות הללו.

לכן אומרים, כי מספר ה-ASCII של האות A הוא 65.

## ASC

גם ללא ספר היצרן יכולת לדעת זאת:

פקוד: ☐

```
PRINT ASC ("A")
```

בתגובה הדפוס המחשב 65, שהוא מספר ה-ASCII של A.

בדוק כך את ה-ASCII של סימנים נוספים. ☐

נסה כעת לבצע את המשימה - הכנס לתא הראשון של המחסנית את האות A, ☐

כלומר את ה-ASCII שלה.

זקוק לרמז? העזר בפקודת POKE.

(תשובה 41)

גם אם פעלת כשורה - דבר לא קרה והאות A לא הודפסה על המסך...

## סמן המחסנית

הסיבה לכך היא שישנו עוד בית בזכרון, ומספרו 198, המשמש כסמן המחסנית. המספר הנמצא בבית זה "אומר" למחשב כמה סימנים עליו לשלוח ממחסנית המקשים.

איזה ערך נמצא עכשיו בבית 198?

(אינך זוכר כיצד לקרוא את הערך הנמצא בבית כלשהו?

ראה יחידה 3, עמוד 13)

(תשובה 42)

זאת גם הסיבה, שכאשר פקדת 65, POKE 631, המחשב לא הדפיס את האות "A":

הערך שנמצא בתא 198 היה 0, המחשב חשב כי המחסנית ריקה, ולכן לא שלף משם את הסימן "A".

נסה שוב לבצע את המשימה - להכניס את "A" כך שהיא תודפס על המסך:

- אל תשכח להכניס את ה-ASCII של A לבית הראשון במחסנית.

- באותה שורה - אמור למחשב כי יש סימן אחד במחסנית.

(תשובה 43)

אם פעלת כשורה המחשב הדפיס A והמסמן נמצא מימינה - כאילו הדפסת

A דרך לוח המקשים!

הכנס שורה זו כשורה 20 "בתוכנית". ☐

הרץ את התוכנית. ☐

אתה רואה כי לאחר ההשהייה המחשב מודיע READY ומדפיס את A.

מה יקרה אם תעביר את שורה 20 לשורה 5, לפני ההשהייה? ☐

לאחר שחשבת -

העבר את שורה 20 ל-5 (מחר את 20) והרץ. ☐

אתה רואה כי לא חל שינוי - המחשב קודם מבצע את ההשהייה ורק אח"כ

מדפיס A. מדוע?

כבר אמרנו - המחשב מתפנה להדפיס את המחסנית רק אחרי גמר הרצת

התוכנית - וזה מה שקרה.

## משימה

הוסף לתוכנית את השורה המוזרה

40 REM

כעת שנה את התוכנית כך, שבסיום ריצתה היא תמחק מן התוכנית את

שורה 40 - בלי מגע יד אדם!



עליך להכניס למחסנית גם את הפעולה RETURN כדי למחוק שורה. העזר  
שוב בטבלת ה-ASCII.

(תשובה 44)

## **משימה נוספת**

שנה את התוכנית כך, שתבצע את השלבים הבאים:


- התוכנית רצה ומבצעת השהייה.
- בסיום ההשהייה – המחסנית מוחקת שורה מתוך התוכנית.
- מיד אח"כ, ללא התערבות מצידך – המחשב ממשיך להריץ את התוכנית ומדפיס את שמך, למשל.

.

**רמז:** בסיום ההשהייה, כדי לגרום למחשב לבצע את הפקודות שבתוך  
המחסנית, עליך לכתוב שורה שתעצור את ריצת התוכנית, המחשב  
יבצע את המחסנית, ואח"כ ידפיס את שמך.  
בתוך המחסנית תהיה, לכן, גם הוראה למחשב ללכת ולהריץ את  
התוכנית החל בשורה המתאימה.

(תשובה 45)

## **חוזרים סוף סוף למשחקי המספרים**

כעת, למעשה, בעזרת מחסנית המקשים, תוכל לפצח את משחק-המספרים  
כך, שהמחשב יגריל באופן חופשי מספרים וסימני-פעולה!  
בסה. 

המשימה אינה קלה!

ננסה לרמוז לך לדרך אפשרית על הפתרון:



הרץ את התוכנית "הטפשית" הבאה:

```
10 PRINT "{CLR/HOME}S=3*4"
20 FOR T=1 TO 2000 : NEXT
30 STOP
40 PRINT S
```

זדאי לא הפתיעה אותך התוצאה:

המחשב מחק את המסך והדפיס בפינה  $S=3*4$ .  
לאחר מספר שניות הוא מודיע BREAK בגלל שורה 30.  
וכעת:

## המשימה

עליך לגרום למחשב לחשב את תוצאת התרגיל, כלומר את ערכו של S - אבל אסור לך להשתמש בפקודה PRINT (וגם לא בקיצור שלה, כמובן).  
(תשובה 46)

## עדיין מתקשה?

הנה עוד רמז:

איזה שינוי עליך להכניס בתוכנית של התרגיל שעשית זה עתה כך, שהפעולה החשבונית לא תופיע על המסך - ובכל זאת תתקבל בסוף התוצאה 12?

(תשובה 47)

ואם בכל זאת עדיין לא התגברת על המכשולים, הנה לפניך שלבי התוכנית:

1. המחשב מגריל 4 מספרים.
2. המחשב מגריל 3 פעולות חשבון מתוך 4 אפשרויות.
3. המחשב מדפיס בפינת המסך את התרגיל, מבלי שפעולות החשבון יראו על המסך (כמו בתרגיל שעשית).

4. בעזרת המחסנית המחשב מבצע את פעולות החשבון השונות.
5. המחשב בודק אם הערך של המשתנה הוא מספר שלם. אם לא - חוזר המחשב ומגריל מספרים ופעולות מהתחלה.
6. אם הערך הוא מספר שלם - המחשב מציג את המספרים ואת התוצאה - ועליך לפתור את החידה.
7. בלחיצת מקש נוספת - הוא מציג לפניך את התשובה.

(תשובה 48)

## לסיום

בפרק זה חשפנו את סוד פעולתה של מחסנית המקשים. הראינו לך כיצד ניתן לנצל בצורה מתוככמת את המחסנית לצורך שיפור "משחק-המספרים".

למעשה, אפשר להשתמש במחסנית המקשים לצורך פעולות נוספות במחשב:

- טעינה של תוכנית, הרצתה ובאופן אוטומטי טעינה של תוכנית נוספת והרצתה.
  - מחיקת שורות מתוך תוכנית.
- אפשר גם לבטל את המחסנית (POKE 198, 0) כדי לנקות את המחשב מלחיצות מקשים קודמות, שאינן רצויות.

# פרק ה


## מוט ההגוי – JOYSTICK

### מבוא

אנו מתארים לעצמנו כי ברשותך עומד לפחות מוט היגוי – JOYSTICK (ג'ויסטיק) אחד. בפרק זה תלמד כיצד לתכנת את הג'ויסטיק ולשלב אותו בתוך התוכניו שתכתוב. תוכל מיד להשתמש בידע שתצבור על מנת לשלב את הג'ויסטיק בכל התוכניות, אשר בהם שלטת על התנועה במסך בעזרת מקשי המסמן (ראה חוברת 2 פרק ו, חוברת 3 פרק ו וחוברת 4 פרק ה). בהמשך תלמד כיצד לשלב את "ההדק" של הג'ויסטיק בתוכניות. תוך כך לימוד הפרק נצל את שתי התוכנות של מחשבת "HIGH RESOLUTION" ו-"SPRITES" העומדות כבר לרשותך.

### חקירה ראשונית

חבר את הג'ויסטיק ל-CONTROL PORT 2 שבצד ימין של המחשב (החיבור הרחוק ממך יותר):



עליך לחבר את קצה החוט של מוט ההיגוי לכאן

### משימה

כתוב תוכנית שתדפיס ללא הרף, במקום קבוע על המסך, את הערך הנמצא בבית 56320.

## תזכורת

ביחידה 3 פרק א' למדת, כי זכרוננו של המחשב בנוי ממספר רב של **תאי זכרון**.

כל תא בו **ממוספר**. מספר זה הוא **הכתובת הקבועה** של תא הזכרון.

כל תא זכרון כזה נקרא בשם **BYTE – בֵּיט**

על מנת לקרוא את הערך הנמצא בבית מסוים עליך להשתמש בפקודה \_\_\_\_\_  
(אינך זוכר? ראה יחידה 3 עמוד 13).

ועכשיו השלם את המשימה הקטנה שהתבקשת לבצע.

(תשובה 49)

הרץ את התוכנית ושחק עם הג'ויסטיק – הטה אותו לכיוונים שונים ולחץ גם על כפתור הירי.


אתה רואה כי לבית 56320 יש תכונה מיוחדת:

הוא מתעדכן כל הזמן באופן שוטף לגבי מצבו של מוט ההיגוי. כך לדוגמא כאשר אינך נוגע במוט ההיגוי ובהדק, המספר הנמצא בבית 56320 הוא: \_\_\_\_\_.

כאשר אתה מטה את המוט למעלה הבית מתעדכן ומקבל את הערך \_\_\_\_\_.

כתוב את הערכים שבית 56320 מקבל כאשר מטים את מוט ההיגוי בכיווני החיצים:

(ללא לחיצה על ההדק)



(תשובה 50)

נצל את הידע שצברת זה עתה על מנת לפתח "עיפרון וידאו עדין", אשר יצייר גם קווים באלכסון בהתאם לכיוון ההטיה של מוט ההיגוי (ראה יחידה 4 פרק ה).  
אל תשכח להשתמש בתוכנה HIGH RESOLUTION של מחשבת.

## משחק: ריצה למרחקים קצרים

אם למדת את מחשבת 6 (ספרייט), ובמצאת בידך התוכנה "SPRITE" של מחשבת, תוכל לתכנת את המשחק הבא:


- תכנת שוב (או העתק - אם הקלטת) את אי-ציק-הולך-להוליבוד.
- תכנת אותו כך שהוא מתקדם רק כאשר אתה מניע את הג'ויסטיק מצד לצד (שמאלה-ימינה).

כעת תוכל לבצע בגד חברך (או בגד המחשב) תחרות ריצה באולימפיאדה מי יגיע ראשון לקו הגמר.  
תוכל, כמובן, להתחרות נגד המחשב או בגד השעון.

## ההדק

אילו ערכים מקבל בית 56320 כאשר בנוסף להטיה אתה לוחץ גם על ההדק?

רשום את הערכים השונים כאן:



(תשובה 51)

האם אתה זוכר את המשחק "יורים ברעים" שפיתחת ביחידה 2?  
(תוכנית המשחק במצאת ביחידה 2 עמוד 65)



שורה 60 באותה תוכנית היא "ההדק". איזו שורה היית כותב במקום

שורה 60 אם ברצונך להשתמש בהדק של הג'ויסטיק למשחק.

(תשובה 52)

## PORT 1


בית 56320 מתעדכן באופן רצוף לגבי מצבו של הג'ויסטיק המחובר לכניסה 2 (PORT 2).



איזה בית בזכרון מתעדכן באופן רצוף במקרה והג'ויסטיק מחובר לכניסה 1? (PORT 1)

ובכן, זה פשוט הבית השכן, כלומר: 56321

חקור אילו ערכים מקבל בית 56321 כאשר הג'ויסטיק מחובר ל-PORT 1.




אתה רואה כי בית 56321 מקבל

ערכים שונים מאלו של בית 56320.

לא בופתע אם אתה מגלה אכזבת-מה מהעובדה, ששתי הכניסות של הג'ויסטיק מקבלות ערכים שונים.

מיד נציע לך כלי תכנותי, אשר יאפשר לך להתגבר על השוני שבין שתי הכניסות.



הרץ את התוכנית הבאה:



10 N=15-PEEK (56321) AND 15

20 PRINT N

30 GO TO 10

אילו ערכים מקבל N במצבי ההטיה השונים?




(שים לב כי הג'ויסטיק מחובר לכניסה 1)

ו? ואילו ערכים מקבל N אם הג'ויסטיק

מחובר לכניסה 2 (PORT 2)?

(אל תשכח לשנות את מספר הבית  
בפקודת PEEK).



## קופסה שחורה

אתה בודאי שואל את עצמך מה סודה של השורה


$10 N=15-PEEK ( \quad ) \text{ AND } 15$


↑  
מספר הבית אליו מחובר הג'ויסטיק


מאחר שסוד פעולתה של שורה זאת נעוץ בהבנה מעמיקה יותר של המבנה הפנימי של הבתים (BYTES), לא ננסה בשלב זה להסבירם. בינתיים נתיחס לשורה זאת כאל "קופסה שחורה". כלומר, אינך יודע כיצד בדיוק היא פועלת - אבל אתה יודע כי היא מאפשרת לך לקבל התאמה בין שתי הכניסות של הג'ויסטיק. למעשה בחיי יום יום, אתה משתמש בהרבה "קופסאות שחורות" - אתה יודע להשתמש בטלויזיה - אך האם אתה יודע בדיוק את אופן פעולתה?


ביחידה הבאה שתוקדש כולה ללימוד מעמיק יותר של הזכרון ומבנה הבתים, תרכוש את הידע הנדרש, שיחשוף את סוד פעולתה של שורה זאת, המהווה לעת עתה קופסה שחורה.



האם ערכי N משתנים אם תלחץ על ההדק תוך כדי ההטייה? 

נסה. 


כיצד, אם כך, תשתמש בהדק? 

הוסף עוד שורה לתוכנית כך, שבלחיצה על ההדק, גם תוך כדי הטייה, המחשב ידפיס שורה של כוכביות, למשל. 

(תשובה 53)

## מחק אלגנטי לעפרון וידאו

ביחידה 4 פרק ה' עסקנו בעיפרון וידיאו עדין.

השתמש בעובדה שערכי N אינם מושפעים מן ההדק של הג'ויסטיק ובנה  לך "מחק-אלגנטי" לעיפרון הוידאו, כלומר תוכל להתקדם עם העיפרון מבלי לצייר על המסך, אם ההדק לחוץ.

## עפרון וידאו מהיר

עיפרון הוידאו ביחידה 4 היה די איטי, גם אחרי השימוש בביטויים לוגיים.

אם תעזר ב:

- הקופסה השחורה.

- הפקודה ... GO TO ... ON שלמדת בפרק משחקי-מספרים.

תוכל לתכנת עיפרון וידיאו מהיר יותר, הכולל תנועה אלכסונית עם ציור קו ובלעדי (מחק).

(תשובה 54)

בפרק הבא תנצל את תוכנית העיפרון כדי לירות בצלחות חרס.

באחת היחידות הבאות של "מחשבת" נעסוק בשפת מכונה ואז העיפרון שלך יהיה מהיר באמת...

# פרק ו

## ירי בצלחות חרס


### מבוא

עכשיו, לאחר שלמדת כיצד ניתן לשלב את הג'ויסטיק בתוכניות שלך נציע לך לסיום יחידת הלימוד לפתח את המשחק "ירי בצלחות חרס". שים לב! כדי לפתח את המשחק עליך להשתמש בתוכנה "SPRITES" של מחשבת.

### תאור המשחק


1

מיד לאחר הרצת התוכנית מופיעה על המסך כוונת קטנה. ניתן להזיז את הכוונת לכל כיוון.



2

מיד בהמשך מופיעה צלחת חרס אשר עפה במסלול עקום על המסך. אם הצלחת לא תיפגע היא תעלם מצדו הימני של המסך.



**מטרתך** לפגוע בצלחת החרס בעזרת הכוונת.

בידך כל הכלים התכנותיים לביצוע המשימה...

## רמז 1

הכוונת וצלחת החרס יהיו בנויים כל אחד מספרייט.

## רמז 2

זוכר את עיפרון הוידאו המהיר שפיתחת בסיום הפרק הקודם?  
הזז את הכוונת על המסך באותו אופן.

## רמז 3

מסלול צלחת החרס הוא עקום.  
הוא בודאי מזכיר לך קטע של סינוסואידה.  
(אינך זוכר מהי סינוסואידה? ראה יחידה 5 פרק ג.)  
עליך לגרום לצלחת החרס לנוע על קטע של סינוסואידה.

## רמז 4

ברגע שאתה לוחץ על ההדק המחשב בודק אם הכוונת אכן נמצאה על  
"צלחת החרס". לשם כך עליך להעזר ב-COLLIDE של הספרייט. אם אינך  
זוכר את השימוש ב-COLLIDE פנה ליחידה SPRITES (יחידה 6) עמוד 31.  
אם פגעת - המחשב מודיע "BOOL".  
אם לא - המחשב מודיע "PHYSPHUS".  
(עכשיו תוכל להשוות את פתרוןך עם תשובה 55)

## ומסיימים בשכלולים

הנה כמה הצעות לשכלולים שאתה יכול להכניס:  
1. עליך לפגוע בצלחת בזמן קצר ככל האפשר מרגע הופעתה על המסך.  
ככל שהזמן קצר יותר אתה מקבל ניקוד גבוה יותר.

2. המחשב בוחר באופן אקראי בין מספר סוגי מסלולים:  
סינוסואידות שונות, צורות ליסאזו וכדומה.
3. יש באפשרותך לירות 2 פעמים.  
אם אתה פוגע ביריה הראשונה אתה מקבל ניקוד גבוה יותר.
4. לרשותך 10 כדורים. נראה בכמה מהם תצליח לפגוע בצלחת עד  
שהיא גומרת את המסלול - או עד שנגמרים לך הכדורים...

# תשובות

## תשובה 1

```
GRAPHICS : CLS : PLOT 10, 10
```

## תשובה 2


```
PLOT OVER 1; 10, 10
```

## תשובה 3

```
10 GRAPHICS : CLS
```

```
20 PRINT AT 0, 24; "A"
```

## תשובה 4



## תשובה 5

```
10 GRAPHICS : CLS
```

```
20 PRINT AT 0, 24; "A"
```

```
30 FOR Y=0 TO 7
```

```
40 FOR X=0 TO 7
```

```
50 IF PIXEL (X,Y)=1 THEN:PRINT AT 15-Y; !%-Y; " "
```

```
60 NEXT X, Y
```

## תשובה 6

```
10 GRAPHICS : CLS
```

```
20 GET A$ : IF A$="" THEN 20
```

```
30 CLS
```

```
40 PRINT AT 0, 24; A$
```

```
50 FOR Y=0 TO 7
```

```
60 FOR X=0 TO 7
```

```
70 IF PIXEL (X,Y)=1 THEN : PRINT AT 15+X, 15-Y; " "
```

```
80 NEXT X,Y
```

```
90 GO TO 20
```

## תשובה 7

70 IF PIXEL (X,Y)=1 THEN : PRINT AT 15+X, 15-Y; A\$

## תשובה 8

```
10 A=0 : B=50
20 GRAPHICS : CLS
30 PRINT AT 0, 24; "A"
40 FOR Y=0 TO 7
50 FOR X=0 TO 7
60 IF PIXEL (X,Y)=1 THEN : PLOT A, B
70 A=A+1
80 NEXT X
90 A=0 : B=B+1
100 NEXT Y
```

## תשובה 9

```
60 IF PIXEL (X,Y)=1 THEN : PLOT A, B : PLOT A+1, B
70 A=A+2
```

## תשובה 10

יש להוסיף או לשנות:

```
25 PRINT AT 1, 1; "A$=" : INPUT A$
27 PRINT AT 1, 1; "MX=" : INPUT MX
30 PRINT AT 0, 24; A$
205 FOR I=0 TO MX
210 PLOT A+I, B
220 NEXT I
```

## תשובה 11

יש לשנות:

```
205 למחוק
210 PLOT A, B
220 DRAW MX, 0
```

כדי להגביה את האות פי 2 יש להוסיף או לשנות:

22Ø DRAW MX-1, Ø

9Ø A=Ø : B = B+2

23Ø PLOT A, B+1

24Ø DRAW MX-1, Ø

1Ø A=Ø : B=5Ø

2Ø GRAPHICS : CLS

25 PRINT AT 1, 1; "A\$=" : INPUT A\$

27 PRINT AT 1, 1; "MX=" : INPUT MX

28 PRINT AT 1, 1; "MY=" : INPUT MY

29 CLS

3Ø PRINT AT Ø, 24; A\$

35 REM SCANNING THE ORIGINAL CHARACTER

4Ø FOR Y=Ø TO 7

5Ø FOR X=Ø TO 7

6Ø IF PIXEL (X,Y)=1 THEN GOSUB 2ØØ

7Ø A=A+MX

8Ø NEXT X

9Ø A=Ø : B=B+MY

1ØØ NEXT Y

11Ø GO TO 25

19Ø REM \*\*\*\*\*

2ØØ REM "PRINTING THE BIG CHARACTER"

21Ø FOR I=Ø TO MY

23Ø PLOT A, B+I

24Ø DRAW MX-1, Ø

245 NEXT I

25Ø RETURN

26Ø REM \*\*\*\*\*

## תשובה 15

210 FOR I=0 TO MY STEP 2

יש לשנות:

## תשובה 16

PRINT MID\$ (X\$, 2, 3)

PRINT MID\$ (X\$, 5, 1)

## תשובה 17

300 REM TYPING A STRING BACKWARDS

310 INPUT X\$

320 FOR I=LEN(X\$) TO 1 STEP -1

330 PRINT MID\$ (M\$, I, 1);

340 NEXT I

## תשובה 18

30 FOR J=1 TO LEN (A\$)

יש להוסיף או לשנות

32 CLS

33 PRINT AT 0, 24; MID\$ (A\$, J, 1)

105 A=0 : B=50 : NEXT J

## תשובה 19

המשתנה A צריך לגדול ב:  $8*MX$ .

## תשובה 20

יש להוסיף או לשנות:

32 A=MX\*8\*(J-1) : B=50

90 A=MX\*8\*(J-1) : B=B+MY

105 NEXT J

## תשובה 21

23 PRINT AT 1, 1; "R=" : INPUT R

יש להוסיף או לשנות:

32 A=R\*(J-1)+MX\*8\*(J-1) : B=50

90 A=R\*(J-1)+MX\*8\*(J-1) : B=B+MY

110 GO TO 23



יש להוסיף או לשנות:

```

10 B=10
23 מחוק
27 MX=40/LEN (A$)
28 MY=MX
32 A=R*(J-1)+MX*8*(J-1) : B=10
110 GO TO 25

```

הערה: אם A\$ יהיה אות אחת בלבד, היא תחרוג מחוץ למסך - למעלה.

יש לשנות:

```

60 IF PIXEL (X,Y)=0 THEN GOSUB 200
210 FOR I=0 TO MY

```

יש לשנות:

```

60 IF PIXEL (X,Y)=1 THEN GOSUB 200
210 FOR I=0 TO MY-1 STEP MY/2
240 DRAW MX-1, 0

```

```

30 ON R GO TO 40, 60, 50, 70

```

```

20 R=INT (RND(1)*4)+1

```

```

10 PRINT "{CLR/HOME}"
20 A=INT(RND(1)*9)+1
30 B=INT(RND(1)*9)+1
40 C=INT(RND(1)*9)+1
50 D=INT(RND(1)*9)+1
60 R=INT(RND(1)*3)+1
70 PRINT A;" "; B;" "; C;" "; D : PRINT "{↑1 →}";
80 ON R GO TO 110, 120, 130
110 S=A*B+C-D : PRINT S : GO TO 180
120 S=A+B+C-D : PRINT S : GO TO 180
130 S=A/B+C*D : PRINT S : GO TO 180

```

```

180 GET A$ : IF A$="" THEN 180
200 ON R GO TO 210, 220, 230
210 PRINT "S=A*B+C-D" : GO TO 300
220 PRINT "S=A+B+C-D" : GO TO 300
300 GET A$ : IF A$="" THEN 300
320 GO TO 10

```

```

180 IF INT (S) < S THEN 10
190 GET A$ : IF A$="" THEN 180

```

```

182 IF S<0 THEN 10

```

```

10 DIM W$(30)
20 FOR N=1 TO 30
30 INPUT W$(N)
40 NEXT

```

```

50 DIM W$(30)
60 FOR N=1 TO 30
70 READ W$(N)
80 IF W$(N)="*" THEN N=N-1 : GO TO 200
90 NEXT
1000 DATA ONE, TWO, THREE, FOUR, FIVE, SIX, SEVEN, EIGHT, NINE,
    TEN, *

```

```

200 PRINT " ♥ " : REM GAME 1 *****
210 R=INT (RND(1)*N) + 1
220 PRINT W$ (R)
230 FOR T=1 TO 300 : NEXT : PRINT "{CLR/HOME}"
240 INPUT A$
250 IF A$=W$(R) THEN 200
260 PRINT "WRONG !!!"
270 FOR T=1 TO 1000:NEXT:PRINT "{CLR/HOME}"
280 GO TO 220

```

## תשובה 34

```

80 IF W$(N) = "*" THEN N=N-1 : GO TO 100
100 REM MENU
110 PRINT "{CLR/HOME +2 → }";
120 PRINT "M E N U"
130 PRINT "{+3 → }1. GAME 1"
140 PRINT "{+1 → }2. GAME 2"
150 PRINT "{+1 → }3. GAME 3"
160 PRINT "{+1 → }4. GAME 4"
170 INPUT A
180 ON A GO TO 200, 400, 600, 800

```

## תשובה 35

```

400 REM GAME 2 *****
410 PRINT "{CLR/HOME +2 → }";
420 R=INT (RND(1)*N)+1
430 P=INT (RND(1)*LEN(W$(R)))+1
450 IF I=P THEN PRINT "-"; : GO TO 470
460 PRINT MID$ (W$(R),I,1);
...המשך בעמוד הבא...

```

470 NEXT I

475 PRINT

480 INPUT A\$

490 IF A\$=W\$(R) THEN PRINT "VERY GOOD!" : FOR T=1 TO 1000 :

NEXT : GO TO 400

500 PRINT "WRONG!" : FOR T=1 TO 1000 : NEXT : PRINT "{CLR/HOME

+2 →}" : GO TO 440

תשובה 36

255 IF A\$ = "&lt;" THEN 100

יש להוסיף:

תשובה 37

600 REM GAME 3 \*\*\*\*\*

610 PRINT "{CLR/HOME +2 →}";

620 R=INT (RND(1)\*N)+1

630 FOR I=1 TO LEN(W\$(R))

640 PRINT "-";

650 NEXT I

660 GET A\$ : IF A\$="" THEN 660

670 IF A\$="&lt;" THEN 100

675 IF A\$="=" THEN 600

680 PRINT "{HOME +2 →}";

690 FOR I=1 TO LEN (W\$(R))

700 IF A\$=MID\$ (W\$(R),I,1) THEN PRINT A\$; : GO TO 720

710 PRINT "{→}";

720 NEXT I

750 GO TO 660

202 PRINT "1. FAST" : PRINT

תשובה 38

204 PRINT "2. MEDIUM" : PRINT

206 PRINT "3. SLOW" : PRINT

208 GET A : IF A=0 THEN 208

209 PRINT "{CLR/HOME}"

210 R=INT (RND(1)\*N)+1

220 PRINT W\$ (R)

230 FOR T=1 TO 200\*A:NEXT:PRINT "{CLR/HOME}"

### תשובה 39

במחשנית יש מקום ל-10 סימנים.

### תשובה 40

ה-ASCII של "A" הוא 65.

### תשובה 41

POKE 631, 65

### תשובה 42

PRINT PEEK (198)

### תשובה 43

POKE 631, 65 : POKE 198, 1

### תשובה 44

5 POKE 631, ASC ("4") : POKE 632, ASC ("0") :  
POKE 633, 13 : POKE 198, 3

### תשובה 45

```
5 FOR T=1 TO 1000 : NEXT
10 POKE 631, ASC ("5") : POKE 632, ASC ("5") : POKE 633, 13
15 POKE 634, ASC ("G")
20 POKE 635, ASC ("O")
25 POKE 636, ASC ("T")
30 POKE 637, ASC ("O")
35 POKE 638, ASC ("6")
45 POKE 640, 13 ("0")
50 POKE 198, 10
55 REM
60 PRINT "DANI"
```

- מחזירים את המסמן "הביתה" (פינה שמאלית עליונה).
- לוחצים RETURN.
- פוקדים GO TO 70

יש לשנות: "{CLR/HOME}{צבע כחול}S=3\*4{לבן}"  
 כלומר, לכתוב את התרגיל בצבע המסך, כך שלא יראה.

```

40 FOR I=1 TO 4
50 N(I)=INT (RND(1)*9)+1
55 R(I)=INT (RND(1)*4)+1
60 NEXT
70 A$(1)="+"
72 A$(2)="-"
74 A$(3)="*"
76 A$(4)="/"
80 PRINT "{CLR/HOME}{כחול}S=";
90 FOR I=1 TO 3
100 PRINT N(I); A$(R(I));
110 NEXT
120 PRINT N(4)
140 POKE 631, ASC ("HOME")
150 POKE 632, 13
160 POKE 633, ASC ("G")
170 POKE 634, ASC ("O")
180 POKE 635, ASC ("T")
190 POKE 636, ASC ("O")
200 POKE 736, ASC ("2")

```

המשך התשובה בעמוד הבא...

```

210 POKE 638, ASC ("6")
220 POKE 639, ASC ("0")
230 POKE 640, 13
240 POKE 198, 10
250 STOP
260 IF S<>INT (S) OR S<0 OR S>250 THEN RUN
300 PRINT "{לבן}{CLR/HOME +28}";
310 PRINT N(1); N(2); N(3); N(4)
320 PRINT "{+112}"; S
330 GET A$ : IF A$="" THEN 330
335 PRINT "{+44}";
340 FOR I=1 TO 3
350 PRINT N(I); A$(R(I));
360 NEXT
370 PRINT N(4); "="; S

```


תשובה 49

```


5 PRINT "{CLR/HOME}"
10 PRINT "{HOME} +610}"; PEEK 56320
20 GO TO 10

```

תשובה 50



320/58  
port 2



3.11.2

תשובה 52

60 IF PEEK 56320=111 THEN 200

תשובה 53

25 IF PEEK (56320) < 112 THEN PRINT "\*\*\*\*\*"

יש להוסיף:

תשובה 54

```

10 GRAPHICS : CLS
20 X=150 : Y=100
30 N=15-PEEK (56320) AND 15
40 ON N GO TO 110, 120, 130, 140, 150, 160, 10, 180, 190, 200
50 GO TO 300
110 Y=Y+1 : GO TO 300
120 Y=Y-1 : GO TO 300
140 X=X-1 : GO TO 300
150 X=X-1 : Y=Y+1 : GO TO 300
160 X=X-1 : Y=Y-1 : GO TO 300
180 X=X+1 : GO TO 300
190 X=X+1 : Y=Y+1 : GO TO 300
200 X=X+1 : Y=Y-1 : GO TO 300
300 PLOT X, Y
500 PLOT OVER (-(PEEK(56320)<112)); X, Y
700 GO TO 30
    
```



```

5 H=200 : V=200
10 PRINT "{CLR/HOME}"
20 FOR X=1 TO 400 STEP 2
25 Y=200-150*SIN(PI*X/320)
30 SPRITE 2, XX, YY, S1, C1
40 SPRITE 1, XH, YV, S2, EX0, EY0
50 N=15-PEEK(56320) AND 15
55 Z= COLLIDE(2)
60 IF PEEK(56320) < 112 THEN 600
70 ON N GO TO 200,210,220,230,240,250,260,270,280,290
150 NEXT
200 V=V-3 : GO TO 150
210 V=V+3 : GO TO 150
230 H=H-3 : GO TO 150
240 H=H-3 : V=V-3 : GO TO 150
250 H=H-3 : V=V+3 : GO TO 150
270 H=H+3 : GO TO 150
280 H=H+3 : V=V-3 : GO TO 150
290 H=H+3 : V=V+3 : GO TO 150
600 IF COLLIDE(2)=2 THEN PRINT "BOOL" : STOP
610 PRINT "PHISPHUS"

```

הערה:

אנחנו בנינו את הצלחת בעורך הספרייטים מריבוע של 4 משכצות בלבד  
והכוונת היא צלב בגודל 5×5.

## רשימת מושגים ולמדים

לפניך רשימת מושגים שלמדת ביחידה זו.  
עבור על כל אחד מהם ובדוק אם אתה זוכר אותם.  
(תוכל לרענן את זכרוןך בעזרת מספרי העמודים המופיעים  
בסוגריים ליד כל מושג):

- (6) PIXEL
- (16) MID\$
- (22) ON... GO TO...
- (25) CONT
- (28) מערך מחרוזתי
- (36) מחסנית המקשים
- (37) BYTE
- (39) ASC
- (44) PORT 2
- (47) PORT 1
- (48) קופסה שחורה

## לסיום .

הגעת לסיומה של היחידה השביעית. עכשיו אתה נושא איתך מטען ידיעות רחב ביותר המאפשר לך להכנס גם לתחומים חדשים. ומה בהמשך?

בעתיד הקרוב תצא לאור יחידה 8 שתקרא: "שולטים בזכרון", בה תלמד בפרטי פרטים את המבנה המיוחד של הזכרון של הקומודור 64. ידע זה יאפשר לך, בין השאר, להפוך את הקומודור למכונת כתיבה עברית.

הידע שתרכוש ביחידה זאת מהווה בסיס איתן ללמוד שפת המכונה של הקומודור שגם לה בקדיש בעתיד מספר חוברות.

# עד כה יצאו סדרות "מחשבת" עבור:

● מחשב COMMODORE 64

(3 החוברות הראשונות מתאימות גם ל־VIC-20)

● מחשבי SINCLAIR: ZX-SPECTRUM

ZX-81

● מחשבי APPLE

● מחשבי ATARI

● מחשב SPECTRVIDEO

● מחשב AMSTRAD

● מחשב IBM-PC

● מחשב COMMODORE 128

## יצא לאור "שעשועי מחשבת-1"

"שעשועי מחשבת-1" – מציינת את

תחילתה של סדרה חדשה של

"מחשבת" ומיועדת לכל חובבי

המחשבים הביתיים המהפשיים אתגרים

תכנותיים.

מה תמצא בחוברת הראשונה של

"שעשועי מחשבת"?

הצעות מפורטות לפיתוח 5 משחקי

מחשב המתאימים מאד לתכנות בשפת

BASIC.

החוברת אינה כוללת אף תוכנית – את

זאת אנו משאירים לך לפתח.

בנוסף לכך תוכל להשתתף בהגרלה

נושאת פרסים של "מחשבת-BUG".

(פרטים תמצא בחוברת עצמה).





### יחידה 7 הנקראת "התוכנית הגדולה"

מוקדשת לפיתוח תוכניות מגוונות תוך ניצול  
הידע הרב שצברת בשש החוברות הקודמות.  
שפת BASIC של הקומודור 64 עשירה

ועדיין יש מה ללמוד!

ביחידה זאת תלמד לנצל כלים חדשים

לצורך פיתוח התוכניות, בין השאר:

• תטפל באופן מתוחכם במחרוזות.

• תנצל את מחסנית המקשים לפיתוח משחקי מספרים.

• תנצל את הג'ויסטיק לפיתוח משחקים.

### שים לב!

ביחידת לימוד זאת ננצל את שתי התוכנות:

### HIGH-RESOLUTION SPRITES

המיוחדות של מחשבת אשר שמשו אותך כבר ביחידות הקודמות.